

Computer Organization and Architecture Lab

Verilog Implementation of Adders, Multipliers, and ALU Blocks for Processor Design

Faculty Development Program
IIT Indore

Lab Duration: 2 Hours
Material: Verilog HDL source files and one self-contained lab handout
Prepared by: Mrunal Shende
Department: Computer Science and Engineering, IIT Indore

Lab Theme: From basic arithmetic circuits to an integrated ALU datapath block

Lab Repository: github.com/Mrunal321/COA-Lab

Contents

1	Aim of the Lab	2
2	Accessing the Lab Files	2
3	Background: Processor Datapath Blocks	2
4	Verilog Basics Required for This Lab	3
5	Experiment 1: Half Adder and Full Adder	3
5.1	Half Adder	3
5.2	Full Adder	4
5.3	Testbench for Full Adder	4
6	Experiment 2: 4-bit Ripple Carry Adder	5
6.1	Testbench for 4-bit Ripple Carry Adder	5
7	Experiment 3: 4-bit Multiplier	6
7.1	Testbench for 4-bit Multiplier	6
8	Experiment 4: 4-bit ALU Design	7
8.1	ALU Operation Table	7
8.2	ALU Block Diagram	7
8.3	Verilog Code for 4-bit ALU	7
8.4	Testbench for 4-bit ALU	8
9	Files Provided for the Lab	9
10	Common Debugging Issues	10
11	Mini-Assignment	10

1. Aim of the Lab

Objective

The aim of this lab is to implement basic arithmetic and logic blocks used in processor datapaths using Verilog HDL. The lab begins with half adder and full adder circuits, builds a 4-bit ripple carry adder, implements a 4-bit multiplier, and finally integrates arithmetic and logical operations into a 4-bit ALU.

Learning Outcomes

After completing this lab, participants will be able to:

- understand the role of adders, multipliers, and ALUs in processor datapaths;
- write Verilog modules for combinational circuits;
- build larger arithmetic blocks using smaller modules;
- understand how simple testbench inputs are applied over time;
- verify expected outputs using truth tables and Boolean equations.

2. Accessing the Lab Files

Note


All Verilog source files, testbenches, and the handout are available in the lab repository:

github.com/Mrunal321/COA-Lab

Students should open this repository on their systems during the lab session to access the complete set of files.

3. Background: Processor Datapath Blocks

A processor datapath contains the hardware blocks that perform arithmetic, logic, data movement, and storage operations. In a basic processor, the ALU is one of the central datapath components. It receives operands from registers, performs an operation selected by a control signal, and produces a result.



Main Components Covered in This Lab

- **Adder:** Performs binary addition.
- **Multiplier:** Performs binary multiplication.
- **ALU:** Selects and performs arithmetic or logical operations based on an opcode.

4. Verilog Basics Required for This Lab

A Verilog design is written as a **module**. A module has input ports, output ports, internal wires, and logic statements. In this lab, design modules are written using **wire** and **assign**, so the hardware structure remains close to the Boolean equations.

```

1 module example_gate(
2     input wire a,
3     input wire b,
4     output wire y
5 );
6
7 assign y = a & b;
8
9 endmodule

```

Listing 1: Basic Verilog module template

Important Verilog Constructs

- **assign**: used for continuous assignment in combinational logic.
- **wire**: used for connections between logic blocks.
- **reg**: used only in the testbench here, because testbench input values change with time.
- **#10**: used in the testbench to wait for 10 ns before applying the next input.

5. Experiment 1: Half Adder and Full Adder

5.1 Half Adder

A half adder adds two one-bit inputs A and B . It produces a sum bit and a carry bit.

$$Sum = A \oplus B = A'B + AB', \quad Carry = A \cdot B$$

A	B	Sum	Carry
0	0	0	0
0	1	1	0
1	0	1	0
1	1	0	1

```

1 `timescale 1ns/1ps
2
3 module half_adder(
4     input wire a,
5     input wire b,
6     output wire sum,
7     output wire carry
8 );
9
10 assign sum    = (~a & b) | (a & ~b);
11 assign carry = a & b;
12
13 endmodule

```

Listing 2: Verilog code for half adder

5.2 Full Adder

A full adder adds three one-bit inputs: A , B , and C_{in} .

$$Sum = A \oplus B \oplus C_{in}$$

$$Sum = A'B'C_{in} + A'BC'_{in} + AB'C'_{in} + ABC_{in}$$

$$Carry = AB + AC_{in} + BC_{in}$$

```

1  `timescale 1ns/1ps
2
3  module full_adder(
4      input wire a,
5      input wire b,
6      input wire cin,
7      output wire sum,
8      output wire cout
9  );
10
11  assign sum = (~a & ~b & cin) |
12              (~a & b & ~cin) |
13              ( a & ~b & ~cin) |
14              ( a & b & cin);
15  assign cout = (a & b) | (b & cin) | (a & cin);
16
17  endmodule

```

Listing 3: Verilog code for full adder

5.3 Testbench for Full Adder

```

1  `timescale 1ns/1ps
2
3  module tb_full_adder;
4
5      reg a;
6      reg b;
7      reg cin;
8      wire sum;
9      wire cout;
10
11      full_adder uut(
12          .a(a),
13          .b(b),
14          .cin(cin),
15          .sum(sum),
16          .cout(cout)
17      );
18
19      initial begin
20          a = 0; b = 0; cin = 0; #10;
21          a = 0; b = 0; cin = 1; #10;
22          a = 0; b = 1; cin = 0; #10;
23          a = 0; b = 1; cin = 1; #10;
24          a = 1; b = 0; cin = 0; #10;
25          a = 1; b = 0; cin = 1; #10;
26          a = 1; b = 1; cin = 0; #10;
27          a = 1; b = 1; cin = 1; #10;
28
29          $finish;
30      end
31
32  endmodule

```


Listing 4: Testbench for full adder

Lab Exercise

For each input combination, use the full adder truth table to verify **sum** and **cout**.

6. Experiment 2: 4-bit Ripple Carry Adder

A ripple carry adder is built by connecting multiple full adders in series. The carry output of one full adder becomes the carry input of the next full adder.



```

1  `timescale 1ns/1ps
2
3  module ripple_carry_adder_4bit(
4      input wire [3:0] a,
5      input wire [3:0] b,
6      input wire      cin,
7      output wire [3:0] sum,
8      output wire      cout
9  );
10
11  wire c1, c2, c3;
12
13  full_adder fa0(
14      .a(a[0]),
15      .b(b[0]),
16      .cin(cin),
17      .sum(sum[0]),
18      .cout(c1)
19  );
20
21  full_adder fa1(
22      .a(a[1]),
23      .b(b[1]),
24      .cin(c1),
25      .sum(sum[1]),
26      .cout(c2)
27  );
28
29  full_adder fa2(
30      .a(a[2]),
31      .b(b[2]),
32      .cin(c2),
33      .sum(sum[2]),
34      .cout(c3)
35  );
36
37  full_adder fa3(
38      .a(a[3]),
39      .b(b[3]),
40      .cin(c3),
41      .sum(sum[3]),
42      .cout(cout)
43  );
44
45  endmodule

```

Listing 5: 4-bit ripple carry adder using full adder modules

6.1 Testbench for 4-bit Ripple Carry Adder

```

1  `timescale 1ns/1ps
2
3  module tb_ripple_carry_adder_4bit;
4
5      reg [3:0] a;
6      reg [3:0] b;

```

```

7  reg      cin;
8  wire [3:0] sum;
9  wire      cout;
10
11 ripple_carry_adder_4bit uut(
12     .a(a),
13     .b(b),
14     .cin(cin),
15     .sum(sum),
16     .cout(cout)
17 );
18
19 initial begin
20     a = 4'd3;  b = 4'd5; cin = 0; #10;
21     a = 4'd7;  b = 4'd8; cin = 0; #10;
22     a = 4'd15; b = 4'd1; cin = 0; #10;
23     a = 4'd10; b = 4'd6; cin = 1; #10;
24
25     $finish;
26 end
27
28 endmodule

```

Listing 6: Testbench for 4-bit ripple carry adder

Discussion Point

The delay of a ripple carry adder depends on carry propagation. The carry must travel from the least significant bit to the most significant bit. This makes the circuit simple, but not the fastest option for large bit-widths.

7. Experiment 3: 4-bit Multiplier

A multiplier produces the product of two binary numbers. For two 4-bit inputs, the maximum value is:

$$15 \times 15 = 225$$

Therefore, the product requires 8 bits.

```

1  `timescale 1ns/1ps
2
3  module multiplier_4bit(
4      input wire [3:0] a,
5      input wire [3:0] b,
6      output wire [7:0] product
7  );
8
9  assign product = a * b;
10
11 endmodule

```

Listing 7: 4-bit combinational multiplier

7.1 Testbench for 4-bit Multiplier

```

1  `timescale 1ns/1ps
2
3  module tb_multiplier_4bit;
4
5  reg [3:0] a;
6  reg [3:0] b;
7  wire [7:0] product;
8
9  multiplier_4bit uut(
10     .a(a),
11     .b(b),
12     .product(product)
13 );
14

```

```

15 initial begin
16     a = 4'd3; b = 4'd4; #10;
17     a = 4'd7; b = 4'd2; #10;
18     a = 4'd15; b = 4'd15; #10;
19     a = 4'd9; b = 4'd6; #10;
20
21     $finish;
22 end
23
24 endmodule

```

Listing 8: Testbench for 4-bit multiplier

Lab Exercise

As an extension, implement a 4-bit array multiplier using AND gates for partial products and adders for summation.


8. Experiment 4: 4-bit ALU Design

An Arithmetic Logic Unit performs arithmetic and logical operations based on a control input called an opcode.

8.1 ALU Operation Table

Opcode	Operation	Description
000	ADD	$A + B$
001	SUB	$A - B$
010	AND	Bitwise AND
011	OR	Bitwise OR
100	Student task 1	Unimplemented in starter code
101	Student task 2	Unimplemented in starter code
110	Student task 3	Unimplemented in starter code
111	Student task 4	Unimplemented in starter code

8.2 ALU Block Diagram



8.3 Verilog Code for 4-bit ALU

Note

This starter ALU is written using structural/dataflow Verilog. It does not use **always**, **case**, or output **reg**. The starter code implements only ADD, SUB, AND, and OR. Opcodes 100–111 are intentionally left unimplemented for the assignment task.


```

1  `timescale 1ns/1ps
2
3  module alu_full_adder(
4      input wire a, b, cin,
5      output wire sum, cout
6  );
7
8  assign sum = (~a & ~b & cin) |
9              (~a & b & ~cin) |
10             (a & ~b & ~cin) |
11             (a & b & cin);
12 assign cout = (a & b) | (b & cin) | (a & cin);
13
14 endmodule
15
16 module alu_4bit(
17     input wire [3:0] a,
18     input wire [3:0] b,
19     input wire [2:0] opcode,
20     output wire [3:0] result,
21     output wire carry,
22     output wire zero
23 );
24
25 wire op2_not, op1_not, op0_not;
26 wire sel_add, sel_sub, sel_and, sel_or;
27
28 wire [3:0] add_result, sub_result;
29 wire [3:0] and_result, or_result;
30
31 wire add_c1, add_c2, add_c3, add_cout;
32 wire sub_c1, sub_c2, sub_c3, sub_cout, sub_borrow;
33
34 assign op2_not = ~opcode[2];
35 assign op1_not = ~opcode[1];
36 assign op0_not = ~opcode[0];
37
38 assign sel_add = op2_not & op1_not & op0_not;
39 assign sel_sub = op2_not & op1_not & opcode[0];
40 assign sel_and = op2_not & opcode[1] & op0_not;
41 assign sel_or = op2_not & opcode[1] & opcode[0];
42
43 alu_full_adder add0(a[0], b[0], 1'b0, add_result[0], add_c1);
44 alu_full_adder add1(a[1], b[1], add_c1, add_result[1], add_c2);
45 alu_full_adder add2(a[2], b[2], add_c2, add_result[2], add_c3);
46 alu_full_adder add3(a[3], b[3], add_c3, add_result[3], add_cout);
47
48 alu_full_adder sub0(a[0], ~b[0], 1'b1, sub_result[0], sub_c1);
49 alu_full_adder sub1(a[1], ~b[1], sub_c1, sub_result[1], sub_c2);
50 alu_full_adder sub2(a[2], ~b[2], sub_c2, sub_result[2], sub_c3);
51 alu_full_adder sub3(a[3], ~b[3], sub_c3, sub_result[3], sub_cout);
52
53 assign sub_borrow = ~sub_cout;
54
55 assign and_result = a & b;
56 assign or_result = a | b;
57
58 assign result = ({4{sel_add}} & add_result) |
59                ({4{sel_sub}} & sub_result) |
60                ({4{sel_and}} & and_result) |
61                ({4{sel_or}} & or_result);
62
63 assign carry = (sel_add & add_cout) |
64                (sel_sub & sub_borrow);
65
66 assign zero = ~(result[0] | result[1] | result[2] | result[3]);
67
68 endmodule

```

Listing 9: Gate-level 4-bit ALU design

8.4 Testbench for 4-bit ALU

```

1  `timescale 1ns/1ps

```

```

2
3 module tb_alu_4bit;
4
5 reg [3:0] a;
6 reg [3:0] b;
7 reg [2:0] opcode;
8 wire [3:0] result;
9 wire      carry;
10 wire      zero;
11
12 alu_4bit uut(
13     .a(a),
14     .b(b),
15     .opcode(opcode),
16     .result(result),
17     .carry(carry),
18     .zero(zero)
19 );
20
21 initial begin
22     a = 4'd7; b = 4'd3;
23
24     opcode = 3'b000; #10;
25     opcode = 3'b001; #10;
26     opcode = 3'b010; #10;
27     opcode = 3'b011; #10;
28
29     a = 4'd15; b = 4'd1;
30     opcode = 3'b000; #10;
31
32     $finish;
33 end
34
35 endmodule

```

Listing 10: Testbench for 4-bit ALU

Discussion Point

The ALU uses the opcode to select one operation from multiple arithmetic and logic operations. In an actual processor, the opcode is generated by the control unit after decoding the instruction.

9. Files Provided for the Lab

Each experiment has a Verilog design file and a small associated testbench file. The testbench only applies input values over time. It does not use display tables or tool-specific commands.

- **Half adder and full adder:** half_adder.v, full_adder.v, tb_half_adder.v, tb_full_adder.v
- **Ripple carry adder:** full_adder.v, ripple_carry_adder_4bit.v
Testbench: tb_ripple_carry_adder_4bit.v
- **Multiplier:** multiplier_4bit.v, tb_multiplier_4bit.v
- **ALU:** alu_4bit.v, tb_alu_4bit.v

10. Common Debugging Issues

Issue	Possible Reason
Output remains unknown x	Input values may not be initialized in the testbench
Compilation error	Module name or port name mismatch
Wrong multiplier output	Product width may be too small
ALU result incorrect	Opcode mapping may be wrong
Carry/borrow not visible	Carry output may not be connected or displayed
Simulation does not stop	Missing <code>\$finish</code> statement

11. Mini-Assignment

Lab Exercise

Complete any two of the unimplemented ALU opcode slots 100–111. You may choose from the following operations, or propose another operation to the instructor:

1. NAND
2. NOR
3. XOR
4. NOT of input A
5. Logical shift left or logical shift right of input A